PHSX 565 Astrophysical Plasma Physics
Problem Set 4 - Accretion Disks
Roy Smart

Prepare Mathematica environment

Clear variables

In[144]:=
```
Clear["Global`*"]
vals = {};
```

Define shortcut to convert rule to equation

In[146]:=
```
r2e = Rule → Equal;
```

Use preprint variable to print derivatives in traditional form

In[147]:=
```
(*$PrePrint=
  #/.Derivative[id__][f_][args__]:→TraditionalForm[HoldForm@D[f[args],#]&[
      Sequence@@(DeleteCases[Transpose[{{args},{id}}],{_,0}]/.{x_,1}:→x)]]&;*)
$PrePrint = TraditionalForm
```

Out[147]= TraditionalForm

Define the del operator

In[148]:=
```
$x = {ϖ, ϕ, z}
∇ /: ∇ f_ := Grad[f, $x, "Cylindrical"]
∇ · f_ := Div[f, $x, "Cylindrical"]
∇ /: ∇×f_ := Curl[f, $x, "Cylindrical"]
∇ · ∇ := Δ
∇ /: ∇^2 := Δ
Δ /: Δ f_ := Laplacian[f, $x, "Cylindrical"]
CenterDot /: (v_ · ∇) f_ := Grad[f, $x, "Cylindrical"].v
```

Out[148]= $\{\varpi, \phi, z\}$

Save the information provided in the problem statement

The average mass of the plasma particle is

In[156]:=
```
m$ = m → mp / 2;
$Assumptions = m > 0 && mp > 0;
```

The dominant, azimuthal flow velocity is given in cylindrical coordinates

as

In[158]:=
```
u$0 = u[ϖ, z] → {0, ϖ Ω[ϖ], 0};
$Assumptions = $Assumptions && ϖ > 0;
```

The gravitational potential is approximated as

In[160]:=
```
ψ$0 = ψ[ϖ] → -(G M)/ϖ + 1/2 Ωk[ϖ]^2 z^2;
$Assumptions = $Assumptions && G > 0 && M > 0 && z ∈ Reals && Ωk[ϖ] > 0;
AppendTo[vals, G → UnitConvert[Quantity["GravitationalConstant"]] ];
```

Where the Keplarian rotation is

In[163]:=
```
Ωk$0 = Ωk[ϖ] → Sqrt[(G M)/ϖ^3] ;
```

Part a.

Write the axial component of the momentum equation using the azimuthal flow velocity and the gravitational potential

The momentum equation in a gravitational potential is given as

In[164]:=
```
M$[ρ_, u_, p_, ψ_] := ρ (∂t u) + ρ ((1/2 (∇ (u.u))) - (u × (∇ × u))) == - (∇ p) - ρ (∇ ψ);
```

Evaluate the z-component of the momentum equation using the information given in the problem statement

In[165]:=
```
(Mz$a = M$[ρ[ϖ, z], u[ϖ, z] /. u$0 , p[ϖ, z], ψ[ϖ] /. ψ$0 /. Ωk$0][[ ;; , 3]]  //
    Simplify) // Framed
```

Out[165]= $\boxed{G M z \rho(\varpi, z) + \varpi^3 p^{(0,1)}(\varpi, z) = 0}$

Find the pressure within the disk

Use the ideal gas law

In[166]:=
```
igl$ = p[ϖ, z] == kB/m ρ[ϖ, z] T[ϖ];
$Assumptions = $Assumptions && kB > 0;
AppendTo[vals, kB → (UnitConvert[Quantity["Boltzmann Constant"]] /.
     "Kelvins" → "KelvinsDifference")];
AppendTo[vals, m → UnitConvert[Quantity["ProtonMass"]]]
```

Out[169]= $\{G \to 6.674\times10^{-11}\ \mathrm{m^3/(kg\,s^2)},\ \mathrm{kB} \to 1.38065\times10^{-23}\ \mathrm{kg\,m^2/(s^2K)},\ m \to 1.6726219\times10^{-27}\ \mathrm{kg}\}$

To write density in terms of pressure and temperature

In[170]:=
```
ρ$a = Solve[igl$, ρ[ϖ, z]][[1, 1]]
```

Out[170]= $\rho(\varpi, z) \to \frac{m\, p(\varpi, z)}{\text{kB}\, T(\varpi)}$

and solve the differential equation found in the first part of this problem

In[171]:=
```
p$a = Mz$a
p$a = p$a /. ρ$a
(p$a = DSolve[{p$a , p[ϖ, 0] == p0[ϖ]}, p[ϖ, z], z][[1, 1]]) // Framed
```

Out[171]= $G\, M\, z\, \rho(\varpi, z) + \varpi^3\, p^{(0,1)}(\varpi, z) = 0$

Out[172]= $\frac{G\, m\, M\, z\, p(\varpi, z)}{\text{kB}\, T(\varpi)} + \varpi^3\, p^{(0,1)}(\varpi, z) = 0$

Out[173]= $\boxed{p(\varpi, z) \to \text{p0}(\varpi)\, e^{-\frac{G\, m\, M\, z^2}{2\, \text{kB}\, \varpi^3\, T(\varpi)}}}$

Show that the vertical scale height is the ratio of the isothermal sound speed to the Keplarian rotation rate.

The isothermal sound speed is the usual

In[174]:=
```
csi$a = csi[ϖ] → √(kB T[ϖ] / m);
$Assumptions = $Assumptions && T[ϖ] > 0 && csi[ϖ] > 0;
```

Write the pressure found in the previous part of the problem in terms of the isothermal sound speed and the Keplarian rotation rate

In[176]:=
```
p1$a = p$a
p1$a = p1$a /. Solve[csi$a /. r2e, T[ϖ]][[1, 1]]
p1$a = p1$a /. Solve[Ωk$0 /. r2e, M][[1, 1]]
```

Out[176]= $p(\varpi, z) \to \text{p0}(\varpi)\, e^{-\frac{G\, m\, M\, z^2}{2\, \text{kB}\, \varpi^3\, T(\varpi)}}$

Out[177]= $p(\varpi, z) \to \text{p0}(\varpi)\, e^{-\frac{G\, M\, z^2}{2\, \varpi^3\, \text{csi}(\varpi)^2}}$

Out[178]= $p(\varpi, z) \to \text{p0}(\varpi)\, e^{-\frac{z^2\, \Omega\text{k}(\varpi)^2}{2\, \text{csi}(\varpi)^2}}$

Now write in terms of the scale height

In[179]:=
```
h$a = h[ϖ] → csi[ϖ] / Ωk[ϖ];
$Assumptions = $Assumptions && h[ϖ] > 0;
```

In[181]:=
```
p2$a = p1$a
(p2$a = p2$a /. Solve[h$a /. r2e, csi[ϖ]][[1, 1]]) // Framed
```

Out[181]= $p(\varpi, z) \to \mathrm{p0}(\varpi)\, e^{-\frac{z^2\, \Omega\mathrm{k}(\varpi)^2}{2\,\mathrm{csi}(\varpi)^2}}$

Out[182]= $\boxed{p(\varpi, z) \to \mathrm{p0}(\varpi)\, e^{-\frac{z^2}{2\, h(\varpi)^2}}}$

Calculate the scale height using example parameters

Save the given parameters to memory

In[183]:=
```
AppendTo[vals, M → 100 Quantity["SolarMass"]];
AppendTo[vals, ϖ → Quantity["AstronomicalUnit"]];
AppendTo[vals, T[ϖ] → 10^4 Quantity["KelvinsDifference"]];
```

Evaluate the scale height

In[186]:=
```
h$a1 = h$a
h$a1 = h$a1 /. csi$a // Simplify
h$a1 = h$a1 /. Ωk$0 // Simplify
(h$a2 = h$a1 /. vals) // Framed
```

Out[186]= $h(\varpi) \to \frac{\mathrm{csi}(\varpi)}{\Omega\mathrm{k}(\varpi)}$

Out[187]= $h(\varpi) \to \frac{\sqrt{\frac{\mathrm{kB}\, T(\varpi)}{m}}}{\Omega\mathrm{k}(\varpi)}$

Out[188]= $h(\varpi) \to \frac{1}{\sqrt{\frac{G\, m\, M}{\mathrm{kB}\, \varpi^3\, T(\varpi)}}}$

Out[189]= $\boxed{h(1\ \mathrm{au}) \to 4.563 \times 10^9\ \mathrm{m}}$

Compare the scale height to the radial coordinate

In[190]:=
```
h[ϖ]/ϖ → (h[ϖ]/ϖ /. h$a /. csi$a /. Ωk$0) /. vals // Framed
```

Out[190]= $\boxed{h(1\ \mathrm{au})\ (1/\mathrm{au}) \to 0.03050}$

So the scale height at 1 AU is only 0.03 AU. This can be considered thin.

Part b.

Write the disk’s surface density in terms of the scale height and the equatorial mass density

Start by using the ideal gas law to find the density from the pressure calculated in Part a.

In[191]:=
```
ρ$b = ρ$a
ρ$b = ρ$b /. p2$a
```

Out[191]= $\rho(\varpi, z) \to \frac{m\, p(\varpi, z)}{\text{kB}\, T(\varpi)}$

Out[192]= $\rho(\varpi, z) \to \frac{m\, \text{p0}(\varpi)\, e^{-\frac{z^2}{2\, h(\varpi)^2}}}{\text{kB}\, T(\varpi)}$

Write in terms of the density at z=0

In[193]:= `ρ$b1 = ρ[ϖ, z] → ρ0[ϖ]` $e^{-\frac{z^2}{2\, h[\varpi]^2}}$`;`

Evaluate the density at z=0 in terms of the pressure at z=0

In[194]:=
```
ρ0$b = Solve[ (ρ[ϖ, z] /. ρ$b) == (ρ[ϖ, z] /. ρ$b1), ρ0[ϖ]][[1, 1]]
```

Out[194]= $\rho 0(\varpi) \to \frac{m\, \text{p0}(\varpi)}{\text{kB}\, T(\varpi)}$

Evaluate the surface density using the given formula

In[195]:= `(Σ$b = Σ[ϖ] →` $\int_{-\infty}^{\infty}$ `ρ[ϖ, z] ⅆz /. ρ$b1) // Framed`

Out[195]= $\boxed{\Sigma(\varpi) \to \sqrt{2\, \pi}\; h(\varpi)\, \rho 0(\varpi)}$

Write the density at z=0 in terms of the surface density

In[196]:=
```
ρ0$b1 = Solve[Σ$b /. r2e, ρ0[ϖ]][[1, 1]]
```

Out[196]= $\rho 0(\varpi) \to \frac{\Sigma(\varpi)}{\sqrt{2\, \pi}\; h(\varpi)}$

Part c.

Write down the radial component of the momentum equation evaluated in the equatorial plane

In[197]:=
```
Mz$c = M$[ρ[ϖ, z], u[ϖ, z] /. u$0 , p[ϖ, z], ψ[ϖ] /. ψ$0 /. Ωk$0] [[ ;; , 1]]
Mz$c = Mz$c /. Solve[Ωk$0 /. r2e, M][[1, 1]] // Simplify
(Mz$c = Mz$c // Simplify) /. z → 0 // Framed
```

Out[197]=
$$\left(\varpi^2 (-\Omega(\varpi))\,\Omega'(\varpi) + \frac{1}{2}\left(2\,\varpi^2\,\Omega(\varpi)\,\Omega'(\varpi) + 2\,\varpi\,\Omega(\varpi)^2\right) - 2\,\varpi\,\Omega(\varpi)^2\right)\rho(\varpi, z) = \rho(\varpi, z)\left(-\left(\frac{G\,M}{\varpi^2} - \frac{3\,G\,M\,z^2}{2\,\varpi^4}\right)\right) - p^{(1,0)}(\varpi, z)$$

Out[198]=
$$\rho(\varpi, z)\left(\left(3\,z^2 - 2\,\varpi^2\right)\Omega k(\varpi)^2 + 2\,\varpi^2\,\Omega(\varpi)^2\right) = 2\,\varpi\,p^{(1,0)}(\varpi, z)$$

Out[199]=
$$\boxed{\rho(\varpi, 0)\left(2\,\varpi^2\,\Omega(\varpi)^2 - 2\,\varpi^2\,\Omega k(\varpi)^2\right) = 2\,\varpi\,p^{(1,0)}(\varpi, 0)}$$

Use scaling arguments to show that the pressure term is negligible

Plug the pressure found in Part a. and the density found in Part b.

In[200]:=
```
p$c = {p :> Function[{ϖ, z}, p[ϖ, z] /. p2$a]};
p0$c = {p :> Function[{ϖ, z}, p0[ϖ] /. p2$a]};
$Assumptions = $Assumptions && p0[ϖ] > 0 ;
Mz$c1 = Mz$c /. Equal → Greater
Mz$c1 = Mz$c1 /. p$c
Mz$c1 = Mz$c1 /. ρ$b
Mz$c1 = Mz$c1 /. z → 0
Mz$c1 = Mz$c1 // Simplify
```

Out[203]=
$$\rho(\varpi, z)\left(\left(3\,z^2 - 2\,\varpi^2\right)\Omega k(\varpi)^2 + 2\,\varpi^2\,\Omega(\varpi)^2\right) > 2\,\varpi\,p^{(1,0)}(\varpi, z)$$

Out[204]=
$$\rho(\varpi, z)\left(\left(3\,z^2 - 2\,\varpi^2\right)\Omega k(\varpi)^2 + 2\,\varpi^2\,\Omega(\varpi)^2\right) > 2\,\varpi\left(\frac{z^2\,\mathrm{p0}(\varpi)\,e^{-\frac{z^2}{2\,h(\varpi)^2}}\,h'(\varpi)}{h(\varpi)^3} + e^{-\frac{z^2}{2\,h(\varpi)^2}}\,\mathrm{p0}'(\varpi)\right)$$

Out[205]=
$$\frac{m\,\mathrm{p0}(\varpi)\,e^{-\frac{z^2}{2\,h(\varpi)^2}}\left(\left(3\,z^2 - 2\,\varpi^2\right)\Omega k(\varpi)^2 + 2\,\varpi^2\,\Omega(\varpi)^2\right)}{\mathrm{kB}\,T(\varpi)} > 2\,\varpi\left(\frac{z^2\,\mathrm{p0}(\varpi)\,e^{-\frac{z^2}{2\,h(\varpi)^2}}\,h'(\varpi)}{h(\varpi)^3} + e^{-\frac{z^2}{2\,h(\varpi)^2}}\,\mathrm{p0}'(\varpi)\right)$$

Out[206]=
$$\frac{m\,\mathrm{p0}(\varpi)\left(2\,\varpi^2\,\Omega(\varpi)^2 - 2\,\varpi^2\,\Omega k(\varpi)^2\right)}{\mathrm{kB}\,T(\varpi)} > 2\,\varpi\,\mathrm{p0}'(\varpi)$$

Out[207]=
$$m\,\varpi\,\mathrm{p0}(\varpi)\left(\Omega(\varpi)^2 - \Omega k(\varpi)^2\right) > \mathrm{kB}\,T(\varpi)\,\mathrm{p0}'(\varpi)$$

Replace derivatives with ratios of the variables

In[208]:=
```
Mz$c2 = Mz$c1 /. D[p0[ϖ], ϖ] → p0[ϖ] / ϖ
Mz$c2 = Mz$c2 // Simplify
Mz$c2 = Mz$c2 /. Solve[h$a1 /. r2e, T[ϖ]] [[1]]
(Mz$c2 = Mz$c2 // Simplify) // Framed
```

Out[208]= $m\,\varpi\,\mathrm{p0}(\varpi)\left(\Omega(\varpi)^2-\Omega\mathrm{k}(\varpi)^2\right)>\frac{\mathrm{kB}\,\mathrm{p0}(\varpi)\,T(\varpi)}{\varpi}$

Out[209]= $m\,\varpi^2\left(\Omega(\varpi)^2-\Omega\mathrm{k}(\varpi)^2\right)>\mathrm{kB}\,T(\varpi)$

Out[210]= $m\,\varpi^2\left(\Omega(\varpi)^2-\Omega\mathrm{k}(\varpi)^2\right)>\frac{G\,m\,M\,h(\varpi)^2}{\varpi^3}$

Out[211]= $\boxed{\varpi^5\left(\Omega(\varpi)^2-\Omega\mathrm{k}(\varpi)^2\right)>G\,M\,h(\varpi)^2}$

Because ϖ << h, the RHS is much smaller than the LHS, so the pressure term is negligible. With the RHS equal to zero, we can show that the azimuthal velocity must have a Keplarian profile

In[212]:=
```
Mz$c3 = Mz$c2 /. h[ϖ] → 0 /. Greater → Equal
Mz$c3 = Mz$c3 // Simplify
(Ω$c = Solve[Mz$c3, Ω[ϖ]][[2, 1]]) // Framed
```

Out[212]= $\varpi^5\left(\Omega(\varpi)^2-\Omega\mathrm{k}(\varpi)^2\right)=0$

Out[213]= $\Omega(\varpi)^2=\Omega\mathrm{k}(\varpi)^2$

Out[214]= $\boxed{\Omega(\varpi)\to\Omega\mathrm{k}(\varpi)}$

Determine if an inwards pressure gradient creates faster or slower rotation

Returning to our expression before the scaling arguments were applied, set the gradient of the pressure to a constant negative value

In[215]:=
```
Mz$c3 = Mz$c1 /. Greater → Equal
Mz$c3 = Mz$c3 /. D[p0[ϖ], ϖ] → -Γ
$Assumptions = $Assumptions && Γ > 0;
Solve[Mz$c3, Ω[ϖ]][[2, 1]] // Framed
```

Out[215]= $m\,\varpi\,\mathrm{p0}(\varpi)\left(\Omega(\varpi)^2-\Omega\mathrm{k}(\varpi)^2\right)=\mathrm{kB}\,T(\varpi)\,\mathrm{p0}'(\varpi)$

Out[216]= $m\,\varpi\,\mathrm{p0}(\varpi)\left(\Omega(\varpi)^2-\Omega\mathrm{k}(\varpi)^2\right)=\Gamma\,(-\mathrm{kB})\,T(\varpi)$

Out[218]= $\boxed{\Omega(\varpi)\to\frac{\sqrt{m\,\varpi\,\mathrm{p0}(\varpi)\,\Omega\mathrm{k}(\varpi)^2-\Gamma\,\mathrm{kB}\,T(\varpi)}}{\sqrt{m}\,\sqrt{\varpi}\,\sqrt{\mathrm{p0}(\varpi)}}}$

From the expression above, we can see that a positive value of Γ (negative pressure gradient) *reduces* the rotation rate

Part d.

Use the continuity equation to show that $\varpi u_\varpi \Sigma$ is a constant

Insert small radial flow component into velocity expression

In[219]:=
```
u$1 = u$0;
u$1[[2, 1]] = uϖ[ϖ];
u$1
```

Out[221]= $u(\varpi, z) \to \{\mathrm{u}\varpi(\varpi), \varpi\,\Omega(\varpi), 0\}$

The continuity equation is often expressed as

In[222]:=
```
C$[ρ_, u_] := (∂t ρ) + (∇ · (ρ u)) == 0
```

Now integrate the continuity equation over all space

In[223]:=
```
IC$d = Integrate[C$[ρ[ϖ, z] /. ρ$b1 /. ρ0$b1 , u[ϖ, z] /. u$1][[1]] ϖ,
    {ϖ, 0, ∞}, {ϕ, 0, 2 π}, {z, -∞, ∞}] == C1
```

Out[223]= $\int_0^\infty 2\pi(\varpi\Sigma(\varpi)\,\mathrm{u}\varpi'(\varpi) + \mathrm{u}\varpi(\varpi)\,(\varpi\Sigma'(\varpi) + \Sigma(\varpi)))\,d\varpi = \mathrm{C1}$

By inspection, we notice that the derivative of the conserved quantity

In[224]:=
```
dC$d = ∂ϖ (ϖ uϖ[ϖ] Σ[ϖ]) // Simplify
```

Out[224]= $\varpi\Sigma(\varpi)\,\mathrm{u}\varpi'(\varpi) + \mathrm{u}\varpi(\varpi)\,(\varpi\Sigma'(\varpi) + \Sigma(\varpi))$

Is equal to the integrand of the form of the continuity equation calculated above

In[225]:=
```
dC$d == IC$d[[1, 1, 3]]
```

Out[225]= True

So by the fundamental theorem of calculus, we may deduce that the quantity $\varpi u_\varpi \Sigma$ is indeed constant.

In[226]:=
```
C$d = ϖ uϖ[ϖ] Σ[ϖ] → C1;
```

Unfortunately, Mathematica cannot handle symbolic integrals very effectively. A more illuminating proof would be to integrate the continuity equation and use the divergence theorem to remove the radial integral.

Compute the mass flux (accretion rate) across a given radius

Use the conserved quantity to compute the fluid velocity

In[227]:=
```
uϖ$d = Solve[C$d /. r2e, uϖ[ϖ]][[1, 1]]
```

Out[227]= $u\varpi(\varpi) \to \frac{C1}{\varpi\,\Sigma(\varpi)}$

The mass flux is then the product of the density and the radial fluid velocity

In[228]:=
```
(M$d = M[ϖ, z] → uϖ[ϖ] ρ[ϖ, z] /. uϖ$d) // Framed
```

Out[228]= $\boxed{M(\varpi, z) \to \frac{C1\,\rho(\varpi, z)}{\varpi\,\Sigma(\varpi)}}$

Part e.

Introduce the radial flow component into the azimuthal momentum equation evaluated in the equatorial plane

Rewrite the momentum equation, eliminating the pressure term and incorporating the viscosity

In[229]:=
```
(*M$1[ρ_,u_,p_,ψ_] := ρ(∂tu) + ρ ((1/2(∇(u.u)))-(u×(∇×u)))==-ρ (∇ ψ) + μ (Δ u) ;*)
M$1[ρ_, u_, p_, ψ_] :=
  ρ (∂t u) + ρ ((1/2 (∇ (u.u))) - (u × (∇ × u))) == -ρ (∇ ψ) + μ ((∇ (∇ · u)) - (∇ × (∇ × u)));
```

Evaluate the azimuthal component of the momentum equation, using information gained from previous subproblems

In[230]:=
```
Ms$0 = Solve[Ωk$0 /. r2e, M][[1, 1]] ;
Mϕ$e = M$1[ρ[ϖ, z], u[ϖ, z] /. u$1 /. Ω[ϖ] → Ωk[ϖ] /. Ωk$0,
    p[ϖ, z], ψ[ϖ] /. ψ$0 /. Ωk$0][[ ;; , 2]] // Simplify
Mϕ$e = Mϕ$e /. Ms$0 /. z → 0
Mϕ$e = Mϕ$e // Simplify
```

Out[231]= $2\,\mathrm{u}\varpi(\varpi)\sqrt{\frac{G M}{\varpi^3}}\,\rho(\varpi, z) + 3\,\mu\sqrt{\frac{G M}{\varpi^5}} = 0$

Out[232]= $2\,\mathrm{u}\varpi(\varpi)\sqrt{\Omega\mathrm{k}(\varpi)^2}\,\rho(\varpi, 0) + 3\,\mu\sqrt{\frac{\Omega\mathrm{k}(\varpi)^2}{\varpi^2}} = 0$

Out[233]= $2\,\varpi\,\mathrm{u}\varpi(\varpi)\,\rho(\varpi, 0) + 3\,\mu = 0$

Solve for the density in terms of the azimuthal flow velocity

In[234]:=
```
ρ$e = Solve[Mϕ$e, ρ[ϖ, 0]][[1, 1]]
```

Out[234]= $\rho(\varpi, 0) \to -\frac{3\,\mu}{2\,\varpi\,\mathrm{u}\varpi(\varpi)}$

Insert this expression into our equation for the accretion rate

In[235]:=
```
(M$e = M$d /. z → 0 /. ρ$e /. uϖ$d) // Framed
```

Out[235]= $\boxed{M(\varpi, 0) \to -\frac{3\,\mu}{2\,\varpi}}$

Evaluate the accretion rate for the point on the disk used in Part a.

Enter the viscosity of an unmagnetized, fully ionized plasma

In[236]:=
```
AppendTo[vals, μ → 0.12 (T[ϖ] / (10^6 Quantity["KelvinsDifference"]))^(5/2)
    Quantity["Grams" / ("Centimeters" "Seconds")]];
```

Evaluate the accretion rate

In[237]:=
```
M$e1 = M$e
M$e1[[2]] = M$e1[[2]] /. vals /. vals;
```

Out[237]= $M(\varpi, 0) \to -\frac{3\,\mu}{2\,\varpi}$

In[239]:=
```
M$e1 /. ϖ → 1 Quantity["AstronomicalUnit"] // Framed
```

Out[239]= $\boxed{M(1\,\mathrm{au}, 0) \to -1.20323\times 10^{-19}\,\mathrm{g/(cm^2 s)}}$

Evaluate the accretion luminosity

Save the value of the speed of light to memory

In[240]:=
```
AppendTo[vals, c → Quantity["SpeedOfLight"]];
```

Evaluate the quantity $M\,c^2$.

In[241]:=
```
Mc$e = M[ϖ, 0] c^2 → (M[ϖ, 0] c^2 /. M$e /. vals /. vals) // UnitConvert // Framed
```

Out[241]= $\boxed{c^2\, M(\varpi, 0) \to -0.108141\ \text{kg/s}^3}$

Compare to the solar luminosity (different units?)

In[242]:=
```
Quantity["SolarLuminosity"] // UnitConvert
```

Out[242]= $3.83\times 10^{26}\ \text{kg}\,\text{m}^2/\text{s}^3$

Part f.

Obtain an expression for the inward radial velocity

We are given that the kinematic turbulent viscosity has the form

In[243]:=
```
νt$f = ν[ϖ] → csi[ϖ] h[ϖ];
```

Kinematic viscosity is related to the dynamic viscosity through

In[244]:=
```
μ$0 = μ → ν[ϖ] ρ[ϖ, z];
```

Evaluate the accretion rate obtained in part e. with this turbulent viscosity coefficient

In[245]:=
```
M$f = M$e
M$f = M$f /. μ$0 /. z → 0
M$f = M$f /. νt$f
```

Out[245]= $M(\varpi, 0) \to -\frac{3\,\mu}{2\,\varpi}$

Out[246]= $M(\varpi, 0) \to -\frac{3\,\nu(\varpi)\,\rho(\varpi, 0)}{2\,\varpi}$

Out[247]= $M(\varpi, 0) \to -\frac{3\,\text{csi}(\varpi)\,h(\varpi)\,\rho(\varpi, 0)}{2\,\varpi}$

Use the definition of the mass flux to solve for the radial velocity

In[248]:=
```
(uϖ$f = Solve[M$f /. M[ϖ, 0] → uϖ[ϖ] ρ[ϖ, 0] /. r2e, uϖ[ϖ]][[1, 1]]) // Framed
```

Out[248]= $\boxed{\mathrm{u}\varpi(\varpi) \to -\frac{3\,\mathrm{csi}(\varpi)\,h(\varpi)}{2\,\varpi}}$

Show for a thin disk that $||u_{\varpi}|| << u_{\phi}$

We can relate the azimuthal velocity to the sound speed through the vertical scale height

In[249]:=
```
Ωk$f = Solve[h$a /. r2e, Ωk[ϖ]][[1, 1]]
```

Out[249]= $\Omega\mathrm{k}(\varpi) \to \frac{\mathrm{csi}(\varpi)}{h(\varpi)}$

Plugging this expression into the azimuthal velocity yields

In[250]:=
```
uϕ$f = uϕ[ϖ] → u$1[[2, 2]] /. Ω → Ωk /. Ωk$f
```

Out[250]= $\mathrm{u}\phi(\varpi) \to \frac{\varpi\,\mathrm{csi}(\varpi)}{h(\varpi)}$

Comparing the above azimuthal velocity to the radial velocity shows

In[251]:=
```
Abs[(uϖ[ϖ] /. uϖ$f)] < (uϕ[ϖ] /. uϕ$f) // Simplify // Framed
```

Out[251]= $\boxed{3\,h(\varpi)^2 < 2\,\varpi^2}$

Since h << ϖ, the azimuthal flow velocity is indeed much larger than the radial velocity

Show for a thin disk that $||u_{\varpi}|| << c_{s,i}$

Comparing the radial flow velocity to the sound speed gives the inequality

In[252]:=
```
(Abs[(uϖ[ϖ] /. uϖ$f)] < csi[ϖ]) // Simplify // Framed
```

Out[252]= $\boxed{3\,h(\varpi) < 2\,\varpi}$

Since h << ϖ, the sound speed is also much larger than the radial velocity

---

Part g.

The problem statement instructs us to consider an isothermal sound speed

In[253]:=
```
csi$g = csi[ϖ] → ξ Sqrt[G M / ϖ];
```

Find the requirements of ξ for a thin disk

Write the above isothermal sound speed in terms of the Keplarian rotation

In[254]:=
```
csi$g1 = csi$g /. Solve[Ωk$0 /. r2e, M][[1, 1]] // Simplify
```

Out[254]= $\mathrm{csi}(\varpi) \to \xi\, \varpi\, \Omega\mathrm{k}(\varpi)$

For a thin disk, h << ϖ, using the definition of h[ϖ] defined in Part a., we have,

In[255]:=
```
e$g1 = (h[ϖ]) < ϖ
e$g2 = e$g1 /. h$a /. csi$g1
e$g2 // FullSimplify // Framed
```

Out[255]= $h(\varpi) < \varpi$

Out[256]= $\xi\, \varpi < \varpi$

Out[257]= $\boxed{\xi < 1}$

So the constant ξ must be much smaller than one and positive for the thin disk limit to hold.

In[258]:=
```
$Assumptions = $Assumptions && ξ > 0 && ξ < 1;
```

Compute the trajectory of the fluid element

The fluid element's trajectory satisfies the equation

In[259]:=
```
T$g = ∂t {ϖ[t], ϕ[t], z[t]} == (u[ϖ, z])
T$g = T$g /. u$1 /. Ω → Ωk
T$g = T$g  /. uϖ$f
T$g = T$g /. h$a
T$g = T$g /. csi$g1
T$g[[2]] = (T$g[[2]] /. Ωk$0 // Simplify) /. ϖ → ϖ[t];
T$g
```

Out[259]= $\{\varpi'(t), \phi'(t), z'(t)\} = u(\varpi, z)$

Out[260]= $\{\varpi'(t), \phi'(t), z'(t)\} = \{\mathrm{u\varpi}(\varpi), \varpi\,\Omega\mathrm{k}(\varpi), 0\}$

Out[261]= $\{\varpi'(t), \phi'(t), z'(t)\} = \left\{-\frac{3\,\mathrm{csi}(\varpi)\,h(\varpi)}{2\,\varpi}, \varpi\,\Omega\mathrm{k}(\varpi), 0\right\}$

Out[262]= $\{\varpi'(t), \phi'(t), z'(t)\} = \left\{-\frac{3\,\mathrm{csi}(\varpi)^2}{2\,\varpi\,\Omega\mathrm{k}(\varpi)}, \varpi\,\Omega\mathrm{k}(\varpi), 0\right\}$

Out[263]= $\{\varpi'(t), \phi'(t), z'(t)\} = \left\{-\frac{3}{2}\xi^2\,\varpi\,\Omega\mathrm{k}(\varpi), \varpi\,\Omega\mathrm{k}(\varpi), 0\right\}$

Out[265]= $\{\varpi'(t), \phi'(t), z'(t)\} = \left\{-\frac{3}{2}\xi^2\sqrt{\frac{G\,M}{\varpi(t)}}, \sqrt{\frac{G\,M}{\varpi(t)}}, 0\right\}$

Solve the above set of differential equations for the trajectory

In[266]:=
```
$Assumptions = $Assumptions && R > 0 && t > 0;
{ϖ$g, ϕ$g, z$g} =
 DSolve[{T$g, ϖ[0] == R, ϕ[0] == 0, z[0] == 0}, {ϖ[t], ϕ[t], z[t]}, t][[2]] //
  Simplify
```

Out[267]= $\left\{\varpi(t) \to \frac{\left(9\,\xi^2\,t\,\sqrt{G\,M} - 4\,R^{3/2}\right)^{2/3}}{2\,\sqrt[3]{2}}, \phi(t) \to \frac{\sqrt[3]{G\,M}\left(18\,\xi^2\,t - \frac{8\,R^{3/2}}{\sqrt{G\,M}}\right)^{2/3} - 4\,R}{6\,\xi^2}, z(t) \to 0\right\}$

Solve for t in terms of the radial coordinate

In[268]:=
```
tϖ$g = Solve[ϖ$g /. ϖ[t] → ϖ /. r2e, t][[1, 1]] // Simplify
```

Out[268]= $t \to \frac{4\left(R^{3/2} + \varpi^{3/2}\right)}{9\,\xi^2\,\sqrt{G\,M}}$

Solve for t in terms of the azimuthal coordinate

In[269]:=
```
tϕ$g = Solve[ϕ$g /. ϕ[t] → ϕ /. r2e, t][[1, 1]] // Simplify
```

Out[269]= $t \to \frac{4\,R^{3/2} + 3\,\xi^2\,\phi\,\sqrt{6\,\xi^2\,\phi + 4\,R} + 2\,R\,\sqrt{6\,\xi^2\,\phi + 4\,R}}{9\,\xi^2\,\sqrt{G\,M}}$

Equate both expressions for t to find the trajectory in the form $\varpi(\phi)$

In[270]:=
```
$Assumptions = $Assumptions && ϕ > 0 && R > 0;
(ϖϕ$g = Solve[(t /. tϖ$g) == (t /. tϕ$g), ϖ][[1, 1]] // FullSimplify) // Framed
```

Out[271]= $\boxed{\varpi \to \frac{3\,\xi^2\,\phi}{2} + R}$

Compute the time required for a fluid element to fall onto the central mass

Use the expression for t[ϖ] calculated in the previous section of the problem to calculate the fall time by setting ϖ=0.

In[272]:=
```
Tf$g = Tf → (t /. tϖ$g /. ϖ → 0)
Tf$g = Tf$g /. Solve[Ωk$0 /. ϖ → R /. r2e, M][[1, 1]]
$Assumptions = $Assumptions && Ωk[R] > 0;
Tf$g = Tf$g // Simplify
(Tf$g = Tf$g /. Ωk[R] → 2 π / Tk[R]) // Framed
```

Out[272]= $\mathrm{Tf} \to \frac{4\,R^{3/2}}{9\,\xi^2\,\sqrt{G\,M}}$

Out[273]= $\mathrm{Tf} \to \frac{4\,R^{3/2}}{9\,\xi^2\,\sqrt{R^3\,\Omega\mathrm{k}(R)^2}}$

Out[275]= $\mathrm{Tf} \to \frac{4}{9\,\xi^2\,\Omega\mathrm{k}(R)}$

Out[276]= $\boxed{\mathrm{Tf} \to \frac{2\,\mathrm{Tk}(R)}{9\,\pi\,\xi^2}}$

Part h.

Solve for acoustic wave modes by using the eikenol approximation with the phase function

In[277]:=
```
φ$h = φ[ϖ, ϕ] → m (f[ϖ] - ϕ);
```

This phase function should satisfy the dispersion relation

In[278]:=
```
ω$h[u_, φ_] := ω == u.(∇ φ) + csa[ϖ] Norm[(∇ φ)]
```

Where the adiabatic sound speed is given by

In[279]:=
```
csa$h = csa[ϖ] → √γ csi[ϖ];
```

In[280]:=
```
$Assumptions = $Assumptions && γ > 0;
```

Solve for the unknown function f[ϖ] for standing wave solutions (ω = 0)

In[281]:=
```
f$h = ω$h[u[ϖ, z] /. u$0, φ[ϖ, ϕ] /. φ$h]
f$h = f$h /. csa$h /. ω → 0
f$h = f$h /. csi$g
f$h = f$h /. Ω → Ωk /. Ωk$0 // Simplify
(f$h = DSolve[f$h, f[ϖ], ϖ][[1, 1]]) /. C[1] → 0 // Framed // Quiet
```

Out[281]= $\omega = \mathrm{csa}(\varpi)\sqrt{\left|m f'(\varpi)\right|^2 + \left|\frac{m}{\varpi}\right|^2} - m\,\Omega(\varpi)$

Out[282]= $0 = \sqrt{\gamma}\,\mathrm{csi}(\varpi)\sqrt{\left|m f'(\varpi)\right|^2 + \left|\frac{m}{\varpi}\right|^2} - m\,\Omega(\varpi)$

Out[283]= $0 = \sqrt{\gamma}\,\xi\sqrt{\frac{G M}{\varpi}}\sqrt{\left|m f'(\varpi)\right|^2 + \left|\frac{m}{\varpi}\right|^2} - m\,\Omega(\varpi)$

Out[284]= $\sqrt{\frac{G M}{\varpi^3}} = \xi\sqrt{\frac{\gamma G M\left(\varpi^2\left|f'(\varpi)\right|^2 + 1\right)}{\varpi^3}}$

Out[285]= $\boxed{f(\varpi) \to -\frac{\sqrt{\frac{1}{\gamma} - \xi^2}\,\log(\varpi)}{\xi}}$

Show that density ridges follow a spiral

Plug the f[ϖ] found in the previous step into the phase function

In[286]:=
```
Solve[{φ$h /. f$h /. r2e /. φ[ϖ, ϕ] → 0}, ϖ][[1, 1]]
```

Out[286]= $\varpi \to e^{\frac{\sqrt{\gamma}\,c_1\,\xi}{\sqrt{1-\gamma\xi^2}} - \frac{\sqrt{\gamma}\,\xi\,\phi}{\sqrt{1-\gamma\xi^2}}}$

This spiral is much tighter than the accretion trajectory since it grows exponentially rather than linearly.